Lecture -February 2

Here for your reading pleasure are the *Maple* calculations for the examples presented in class today--February 1.

First, define and draw a picture of *f*:

```
> f:=x->piecewise(x<1,0,x>1, (x-1));
```

$$f := x \rightarrow \text{piecewise}(x < 1, 0, 1 < x, x - 1)$$

```
> plot(f(x),x=0..2);
```



First, we compute the full Fourier series for *f* :

```
> L:=1;
```

$$L := 1$$

```
> a:=n->(1/L)*int(f(x)*cos((n*Pi*x)/L),x=0..2*L);
```

$$a := n \rightarrow \frac{\int_0^{2L} f(x)\cos\left(\frac{n\pi x}{L}\right)dx}{L}$$

```
> a(n);
```

$$\frac{2\cos(n\pi)^2 - 1 - \cos(n\pi) + 2n\pi\sin(n\pi)\cos(n\pi)}{n^2\pi^2}$$

```
> assume(n,integer);a(n);
```

$$-\frac{-1 + (-1)^{n\sim}}{n\sim^2\pi^2}$$

```
> n:='n';
```

$$n := n$$

```
> b:=n->(1/L)*int(f(x)*sin((n*Pi*x)/L),x=0..2*L);
```

$$b := n \rightarrow \frac{\int_0^{2L} f(x)\sin\left(\frac{n\pi x}{L}\right)dx}{L}$$

```
> assume(n,integer);b(n);
```

$$-\frac{1}{n\sim\pi}$$

```
> n:='n';
```

$$n := n$$

```
> p:=(x,N)->a(0)/2
  +sum((a(n)*cos((n*Pi*x)/L)+b(n)*sin((n*Pi*x)/L)),n=1..N);
```

$$p := (x, N) \to \frac{1}{2}\,a(0) + \left(\sum_{n=1}^{N} \left(a(n)\cos\left(\frac{n\,\pi\,x}{L}\right) + b(n)\sin\left(\frac{n\,\pi\,x}{L}\right)\right)\right)$$

Now here is a picture of the the first 30 terms of the full Fourier series, on the interval [0, 1]:

```
> plot(p(x,30),x=0..2*L);
```



Not too bad!

Now, let's compute the cosine series for the same function f :

```
> L:=2;
```

$$L := 2$$

```
> ac:=n->(2/L)*int(f(x)*cos((n*Pi*x)/L),x=0..L);
```

$$ac := n \to 2\,\frac{\displaystyle\int_0^L f(x)\cos\left(\frac{n\,\pi\,x}{L}\right)dx}{L}$$

```
> assume(n,integer);ac(n);
```

$$-4\,\frac{\cos\left(\frac{1}{2}\,n\sim\,\pi\right) - (-1)^{n\sim}}{n\sim^2\,\pi^2}$$

```
> n:='n';
```

$$n := n$$

```
> pc:=(x,N)->ac(0)/2 + sum(ac(n)*cos((n*Pi*x)/L),n=1..N);
```

$$pc := (x, N) \to \frac{1}{2}\,ac(0) + \left(\sum_{n=1}^{N} ac(n)\cos\left(\frac{n\,\pi\,x}{L}\right)\right)$$

Okay, there it is. Now let's look at a picture of the first 30 terms:

```
> plot(pc(x,30),x=0..2);
```



Pretty good. Next, let's find the sine series:

```
> bs:=n->(2/L)*int(f(x)*sin((n*Pi*x)/L),x=0..L);
```

$$bs := n \rightarrow 2\,\frac{\displaystyle\int_0^L \mathrm{f}(x)\,\sin\left(\frac{n\,\pi\,x}{L}\right)dx}{L}$$

```
> assume(n,integer);bs(n);
```

$$-2\,\frac{(-1)^{n\sim}\,n\sim\,\pi + 2\,\sin\left(\frac{1}{2}\,n\sim\,\pi\right)}{n\sim^2\,\pi^2}$$

```
> n:='n';
```

$$n := n$$

```
> ps:=(x,N)-> sum(bs(n)*sin((n*Pi*x)/L),n=1..N);
```

$$ps := (x, N) \rightarrow \sum_{n=1}^{N} \mathrm{bs}(n)\,\sin\left(\frac{n\,\pi\,x}{L}\right)$$

Again, let's look at a picture:

```
> plot(ps(x,30),x=0..2);
```



Not too bad, but not quite as nice as the others. Can you explain why 30 terms of the sine series doesn't seem to do quite the job 30 terms of the cosine series does?

Now take a look at these three series on a larger interval. Make sure you understand these pictures!

Cosine Series:

```
> plot(pc(x,30),x=-6..6);
```



Sine Series:

```
> plot(ps(x,30),x=-6..6);
```



Full Series:

```
> plot(p(x,30),x=-6..6);
```




```
>
```